
Perl 6 Tutorial: Introduction

The Plan

� I'm going to explain a range of features 
of Perl 6

� Most code examples that I will show in 
this tutorial can be run in Rakudo (Perl 6 
implemented on Parrot) today

� Do ask questions

� If you want to follow along, why not get 
yourself a Rakudo?

Getting Rakudo

� Check out the source

� Configure

� Build

� Get the instructions – takes a while! �

svn co https://svn.perl.org:/parrot/trunk

perl Configure.pl

make
cd languages/perl6 && make

wget http://www.jnthn.net/perl6/build.txt

What We'll Be Covering

� "Hello, world!"

� Variables

� Basic expressions

� Subroutines, parameter passing, etc.

� A range of object oriented features

� Junctions

� Types

� Pattern matching and grammars

Hello, world!

say "Hello, world!";



Perl 6 Tutorial: Hello, world!

say "Salut, les potes!";
say "Comment vas tu yau de poêle";

say "Buenos dias, mundo!";

say "Dobrý de � , svet!";

Variables

Declaring Variables

� As in Perl 5, declare lexical variables 
with my

� Unlike in Perl 5, by default you must 
declare your variables (it's like having 
use strict on by default)

� You can also use our for package 
variables, just like in Perl 5

my $answer = 42;
my $city = 'Bratislava';
my $very_approx_pi = 3.14;

Sigils

� All variables have a sigil

� Unlike in Perl 5, the sigil is just part of 
the name ($a[42] is now @a[42] ).

� Instead, the sigil defines a kind of 
"interface contract" – promises about 
what you can do with this variable

� Anything with @ sigil can be indexed 
into positionally, using […]

Arrays

� Hold zero or more elements and allow 
you to index into them with an integer

# Declare an array.
my @scores;

# Or initialize with some initial values.
my @scores = (52,95,78);
my @scores = <52 95 78>; # The same

# Get and set individual elements.
say @a[1]; # 95
@a[0] = 100;
say @a[0]; # 100

Hashes

� Hold zero or more elements, with keys 
of any type

# Declare a hash.
my %ages;

# Set values.
%ages<Fred> = 19;    # Constant keys
my $name = 'Harry';
%ages{$name} = 23;   # More complex ones 

# Get an individual element.
say %ages<Harry>;    # 23

Everything Is An Object

� You can treat pretty much everything as 
an object if you want

� For example, arrays have an elems
method to get the number of elements

� Can also do push, pop, etc. as methods

my @scores = <52 95 78>;
say @scores.elems; # 3

@scores.push(88);
say @scores.shift; # 52

The for Loop To Iterate

� In Perl 6, the for loop is used to iterate 
over anything that provides an iterator

� By default, puts current value into $_

� The following example will print all of the 
elements in the @scores array

my @scores = <52 95 78>;
for @scores {

say $_;
}

The for Loop To Iterate

� Anything between { … } is just a block

� In Perl 6, a block can take parameters, 
specified using the -> syntax

� Here, we are naming the parameter to 
the block that will hold the iteration 
variable

my @scores = <52 95 78>;
for @scores -> $score {

say $score;
}

The for Loop To Iterate

� .kv method of a hash returns keys and 
values in a list

� A block can take multiple parameters, so 
we can iterate over the keys and values 
together

my %ages = (Jonathan => 25, Fred => 42);
for %ages.kv -> $name, $age {

say "$name is $age years old";
}

The loop Loop

� The for loop is only for iteration now; 
for C-style for loops, use the loop
keyword

� Bare loop block is an infinite loop

loop (my $i = 1; $i <= 42; $i++) {
say $i;

}

loop {
my $cur_pos = get_position();
update_trajectory($target, $cur_pos);

}

The while Loop

� Pretty much as you would expect, 
though like for you don't need the 
parentheses

while !$stomach.full {
my $restaurant = find_restaurant();
my $food = $restaurant.order();
$stomach.eat($food);

} 

The if Statement

� You can use the if…elsif…else style 
construct in Perl 6, as in Perl 5

� However, you can now omit the 
parentheses around the condition

if $foo == 42 {
say "The answer!";

} elsif $foo == 0 {
say "Nothing";

} else {
say "Who knows what";

}

Chained Conditionals

� Perl 6 supports "chaining" of 
conditionals, so instead of writing:

You can just write:

if $role >= 1 && $role <= 6 {
say "Valid dice roll"

}

if 1 <= $role <= 6 {
say "Valid dice roll"

}

Chained Conditionals

� In fact, you are not limited to chaining 
just two conditionals

� Here we check that both roles of the 
dice gave the same value, and that both 
of them are squeezed between 1 and 6, 
inclusive

if 1 <= $role1 == $role2 <= 6 {
say "Doubles!"

}

Declaring And Calling Subroutines

� As in Perl 5, you use the sub keyword to 
declare a subroutine

� Calling it looks the same too

� However, note that & alone won't call 
now – it's just the sub sigil

sub greet {
say "OH HAI!";

}

greet();

Positional Parameters

� Perl 6 makes parameter passing much 
more declarative

� You can still have them all in @_ if you 
really want, but usually you won't 

� Simple positional parameters are just 
listed 

sub greet($who, $greeting) {
say "$greeting, $who!";

}
greet('Pedro', 'Hola'); # Hola, Pedro

Passing Arrays And Hashes

� Unlike in Perl 5, you can just pass two 
arrays without explicitly taking 
references to them and it will Just Work

� Same goes for hashes

sub add_firsts(@a, @b) {
say @a[0] + @b[0];

}

my @a = (1,2,3,4);
my @b = (5,7,8);
add_firsts(@a, @b);  # 6



Perl 6 Tutorial: Subroutines

Optional Parameters

� You can make a parameter optional by 
following it with a ?

� If the value is not supplied, get undef

� You can also supply a default value

sub substr($str, $pos, $length?) {
…

}

sub greet($who, $greeting? = 'Salut') {
say "$greeting, $who!";

}

Named Parameters

� In Perl 5, we often passed "pairs", in the 
form answer => 42

� In Perl 6 the concept is formalized

� The fat arrow operator now creates a 
Pair object; it's not just a comma

� Using these in a parameter list binds 
them to named parameters, which are 
declared differently

Named Parameters

� How to declare named parameters

� You then pass them using pair syntax or 
colonpair syntax

� Notice how order doesn't matter; they 
are bound by the name, not position

sub greet(:$who, :$greeting) {
say "$greeting, $who!";

}

greet(who => 'Pedro', greeting => 'Hola');
greet(:greeting('Hola'), :who('Pedro'));

Named Parameters

� Unlike positional parameters, named 
parameters are optional by default

� Can also give them a default value

� Or you can make them required

sub greet(:$who, :$greeting = 'Salut') {
say "$greeting, $who!";

}

sub greet(:$who!, :$greeting = 'Salut') {
say "$greeting, $who!";

}

Slurpy Parameters

� Sometimes you don't know, ahead of 
time, how many parameters are going to 
be passed or what named parameters 
are going to be passed

� Can get positionals in a slurpy array

� And named in a slurpy hash:
sub total(*@values) { … }

sub total_words(*%word_counts) { … }

Parameter Ordering

� Required positional parameters come 
first

� Optional positional parameters come 
next

� Then all named parameters; the 
ordering of these doesn't matter

� Finally, the slurpy parameters come last

� Can use all of them in one signature



Classes

Classes In Perl 6

� Introduce a class using the class
keyword

� With a block:

� Or without to declare that the rest of 
the file describes the class.

class Puppy {
…

}

class Puppy;

Attributes

� Introduced using the has keyword

� All attributes in Perl 6 are stored in an 
opaque data type

� Hidden to code outside of the class

class Puppy {
has $name;
has $colour;
has @paws;
has $tail;

}

Accessor Methods

� We want to allow outside access to 
some of the attributes

� Writing accessor methods is boring!

� $. means it is automatically generated
class Puppy {

has $ . name;
has $ . colour;
has @paws;
has $tail;

}

Perl 6 Tutorial: Object Orientation



Mutator Methods

� We should be able to change some of 
the attributes

� Use is rw to generate a mutator
method too

class Puppy {
has $.name is rw ;
has $.colour;
has @paws;
has $tail;

}

Attribute Names

� The storage location for an attribute is 
always accessible as $!name

� If you declared it as $name, then 
$name is just an alias to $!name

� If you declared it as $.name , then this 
gets the value through the accessor 
method, whereas $!name refers to the 
underlying storage location

Methods

� The new method keyword is used to 
define a method

� Parameters go in a parameter list; the 
invocant is optional

method bark() {
say “w00f!”;

}

method chew($item) {
$item.damage++;

}

Referring To Attributes In Methods

� Here, we use $!color to get at the 
storage location for the attribute color

� Since there is a mutator, in this case 
we could also have written this as

method play_in_garden() {
$!color = 'black';

}

method play_in_garden() {
$.color = 'black';

}



Getting At The Invocant

� The self keyword can be used to get at 
the current invocant

� You can also put it in a given variable

� Name it in the parameter list separated 
by a colon rather than a comma

self.bark();

method bite($puppy: $unlucky_guy) {
$unlucky_guy.feel_pain;
$unlucky_guy.shout_at($puppy)

}

Using A Class

� A default new method is generated for 
you that sets attributes

my $puppy = Puppy.new(
name => 'Rosey',
colour => 'white'

);
$puppy.bark();           # w00f!
say $puppy.colour;       # white
$puppy.play_in_garden();
say $puppy.colour;       # black

Inheritance

� A puppy is really a dog, so we want to 
implement a Dog class and have Puppy 
inherit from it

� Inheritance is achieved using the is
keyword

class Dog {
…

}
class Puppy is Dog {

…
}

Initializing Parent Attributes

� The default new method is good for 
initializing the attributes of the current 
class

� If we inherited name and color from 
Dog and Puppy has num_teeth, then 
we can do:

my $pup = Puppy.new(
Dog{ name => 'Fido', color => 'white' },
num_teeth => 10

);

Multiple Inheritance

� Multiple inheritance is possible too; use 
multiple is statements

class Puppy is Dog is Pet {
…

}

Roles

In Search Of Greater Re-use

� In Perl 6, roles take on the task of re-
use, leaving classes to deal with 
instance management

� We need to implement a walk method 
for our Dog class

� However, we want to re-use that in the 
Cat and Pony classes too

� What are our options?

The Java, C# Answer

� There’s only single inheritance

� You can write an interface, which 
specifies that a class must implement a 
walk method

� Write a separate class that implements 
the walk method

� You can use delegation (hand coded)

� Sucks

The Multiple Inheritance Answer

� Write a separate class that implements 
the walk method

� Inherit from it to get the method

� Feels wrong linguistically

� “A dog is a walk” – err, no

� “A dog does walk” – what we want

� Multiple inheritance has issues…

Multiple Inheritance Issues

� The diamond inheritance problem

� Do we get two copies of
A’s state?

� If B and C both have a
walk method, which do
we choose?

� Implementing multiple inheritance is 
tricky too

A

B C

D

Mix-ins

� A mix-in is a group of one or more 
methods than can not be instantiated 
on their own

� We take a class and “mix them in” to it

� Essentially, these methods are added 
to the methods of that class

� Write a Walk mixin with the walk
method, mix it in.

How Mix-ins Work

� Defined in terms of single inheritance

� Take C and derive anonymous classes 
with methods of M1, then M2

M1

C

M2 M1

C

M2

����������	

Issues With Mix-ins

� If M1 and M2 both have methods of the 
same name, which one is chosen is 
dependent on the order that we mix in

� Fragile hierarchies problem again

� Further, mix-ins end up overriding a 
method of that name in the class, so 
you can’t decide which mix-in’s method 
to actually call in the class itself

The Heart Of The Problem

� The common theme in our problems is 
the inheritance mechanism

� Need something else in addition

� We want

� To let the class be able to override 
any methods coming from elsewhere

� Explicit detection and resolution of 
conflicting methods

Flattening Composition

� A role, like a mix-in, is a group of 
methods

� If a class does a role, then it will have 
the methods from that role, however:

� If two roles provide the same method, 
it’s an error, unless the class provides 
a method of that name

� Class methods override role methods

Creating Roles

� Roles are declared using the role
keyword

� Methods declared just as in classes
role Walk {

method walk($num_steps) {
for 1..$num_steps {

.step for @paws;
}

}
}

Composing Roles Into A Class

� Roles are composed into a class using 
the does keyword

� Can compose as many roles into a 
class as you want

� Conflict checking done at compile time

� Works? Not quite…

class Dog does Walk {
…

}

Composing Roles Into A Class

� Notice this line in the walk method:

� Can state that a role “shares” an 
attribute with the class it is composed 
into using has without . or !

.step for @paws;

has @paws;

Who likes 
quantum 
physics?

Quantum Superposition

� The theory: if you don’t know what state 
something is in, then it is simultaneously 
in all states until you observe it.

� Example:

� I’m about to explain junctions

� However, you haven’t observed it yet

� Therefore, my explanation is 
simultaneously awful and awesome

One value that is many at the same time

� We know that an array contains many 
values, and a scalar contains a single 
value

� A junction…

� Is a scalar, but…

� Has many values at the same time

� "A quantum superposition of values"

my $scalar = 42; 
my @array = (1, 2, 3);

A simple example

� How often do you find yourself writing 
things like:

� With junctions we can write this as:

� "vodka" | "beer" is a junction

if $drink eq 'vodka' || $drink eq 'beer' {
say "Don't get drunk on it!";

}

if $drink eq 'vodka' | 'beer' {
say "Don't get drunk on it!";

}



Perl 6 Tutorial: Junctions

Another example

� Another example: looping while a 
counter stays below both of two different 
limits:

� Can be re-written as:

while $count < $lim_a && $count < $lim_b {
... 

}

while $count < $lim_a & $lim_b {
... 

}

Constructing Junctions From Arrays

� There are functions to construct 
junctions from arrays too:

if all(@scores) > $pass_mark {
say "Everybody passed!";

}
if $word eq none(@mat) {

say "$word is not swearing!";
}
if any(@scores_a) > all(@scores_b) {

say "Someone in class A beat all of " ~
"the people in class B!";

}

Types Of Junction

� "any" junctions – require one or more of 
the values to match the condition (|)

� "all" junctions require all of the values 
to match the condition (&)

� "one" junctions require exactly one of 
the values to match the condition (^)

� "none" junctions require none of the 
values in the junction to match the 
condition

Auto-threading

� You can use any operator on a junction 
that you would have used with a scalar, 
and the result will be a new junction of 
results.

my $a = 1 | 2;
$a++;               # Now we have 3 | 4
$a = $a + 5;        # Now we have 8 | 9

my $res = "foo" ~ "bar" & "baz";
# "foobar" & "foobaz"

Values Know Their Type

� All values know their types

� Call the .WHAT method to get 
something representing their type

� That something is the proto-object, 
which represents all objects of that type 
and stringifies to the type name

say 42.WHAT;              # Int
say "pivo".WHAT;          # Str
say (answer => 42).WHAT;  # Pair

The Perl 6 View Of Types

� Variables can have a type constraint 
set on them too, which controls what 
can be assigned to them

� Attributes and subroutine parameters 
can also carry type constraints

� Compiler inserts checks to ensure they 
will be checked at runtime at latest 
(compile time good if possible)

Declaring Types

� Variable Declaration

� In subroutine or method parameters

� Can "capture" the type of a parameter 
to implement type-parameterized 
routines too

my Int $foo;

sub trim(Str $s) { … }

sub cmp(::T $a, T $b) { … }

Classes As Types

� You can use the name of a class as a 
type name

� This means that we can only assign 
anything that "isa" Dog to $fido

my Dog $fido;

my Dog $fido;
$fido = Dog.new();     # OK
$fido = Puppy.new();   # OK (Puppy isa Dog)
$fido = Manatee.new(); # Type check error

Manatees Are Not Dogs

isa

Roles As Types

� You can also use the name of a role as 
a type

� This means that we can only assign 
something that "does" the Explode role 
to $ammo

my Explode $ammo;

my Explode $ammo;
$ammo = Grenade.new();  # OK, does Explode
$ammo = Manatee.new();  # Type check error

Manatees Do Not Explode

does

Revisiting Sigils

� Earlier I stated that sigils simply gave 
the promise that what was assigned to 
them met some "interface contract"

� @x requires that whatever is assigned 
to it must do the Positional role

� %x require the Associative role

� &x requires the Callable role

� ::x requires the Abstraction role

Creating Your Own Constraints

� You are not limited to isa and does 
constraints; rather, you can define any 
constraint that you like

� Can be given a name:

� And can be "chained"

subset Even of Int where { $^n % 2 == 0 };
my Even $x;
$x = 200;        # OK
$x = 101;        # Type check error

subset PosEven of Even where { $^n >= 0 };

Dependant Types

� When we parameterize a type of a 
term, it's called a dependant type

� Essentially, types that depend on the 
data

� For example, we can state that the 
head function needs a non-empty list

subset NonEmpty of List
where { $^xs.elems != 0 };

sub head(NonEmpty $xs) { $xs[0] }

Anonymous Constraints

� We don't even need to declare these up 
front

� Here we have a function that combines 
two lists in some way. The second list  
is given a constraint type: it must have 
equal length to the first

sub comb(List @a, List @b
where { @a.elems == @b.elems }) {

...
}

Regexes &
Grammars

Matching

� To match against a pattern, use ~~

� Negated form is !~~

� $/ holds the match object; when used as a 
string, it is the matched text

if $event ~~ /\d**4/ { ... }

if $event !~~ /\d**4/ { say "no year"; }

my $event = "French Perl Workshop 2008"; 
if $event ~~ /\d**4/ {

say "Held in $/"; # Held in 2008
}

The Match Object

� $/ is the source of all match information

� Index into it like an array to get 
positional captures

� You can still use $0, $1 etc as a shortcut to 
this (note they start at 0 now, not 1)

"I love my cat." ~~ /I\s(\w+)\smy\s(\w+)\./;
say $/[0]; # love
say $/[1]; # cat

say $0; # love
say $1; # cat

Named Regexes

� You can now declare a regex with a 
name, just like a sub or method

� Then name it to match against it:

regex Year { \d**4 }; # 4 digits

if $event ~~ Year { ... }

Calling Other Regexes

� You can "call" one regex from another, 
making it easier to build up complex 
patterns.

regex Year { \d**4 };
regex Place { Ukrainian | French | Nordic };
regex Workshop { 

<Place> \s Perl \s Workshop \s <Year>
};
regex YAPC {

'YAPC::' ['EU'|'NA'|'Asia'] \s <Year>
};
regex PerlEvent { <Workshop> | <YAPC> };



Perl 6 Tutorial: Regexes And Grammars

The Match Object

� Using a named rule results in a named 
capture

� Can access them using hash-style 
indexing on the match object

"French Perl Workshop 2008" ~~ PerlEvent;
if $/<YAPC> {

say $/<YAPC><Year>;
} else {

say $/<Workshop><Year>;
}

The Match Object

� There is also abbreviated syntax for 
this

� Note you can make named captures 
without having to have a sub rule

"French Perl Workshop 2008" ~~ PerlEvent;
if $<YAPC> {

say $<YAPC><Year>;
} else {

say $<Workshop><Year>;
}

"YAPC::EU 2008" ~~ /$<Year>=\d**4/

Grammars

� You can collect a bunch or rules 
together in a grammar

� Like a class, but contains regexes 
rather than methods

� Name the main rule TOP
grammar RPNExpression {

token TOP  { ^ <Expr>+ $ };
rule Expr  { <Value>+ <Op>* };
rule Value { \d+[\.\d+]? };
rule Op    { '/' | '*' | '+' | '-' };

}

Grammars

� Grammars are so class-like, and 
named regex inside them so method-
like, that you can actually do 
inheritance

� Perl 6 will ship with its own grammar, 
so you can subclass that to modify the 
syntax => no more source filters!

grammar RPNWithModulo is RPNExpression {
rule Op { '%' | '/' | '*' | '+' | '-' };

}

Merci

Questions?


